
Overview
There are three objectives in this lab:

• understand the mathematical reasoning associated with a real-world example,

• learn to define a piecewise-defined function in Maple, and

• learn to set up and solve a system of equations in Maple.

Maple Essentials
Important Maple commands introduced in this lab are:

Command Description
piecewise define a piecewise-defined function

The general syntax to represent




f1, cond1

f2, cond2

...
...

fn, condn

is:

piecewise( cond1, f1, cond2, f2, ..., condn, fn );
where each condi is an inequality and each fi is an expression.

Maple does not recognize double inequalities, so if your condition is a ≤ x < b you would write x>=a and x<b.

Preparation
Review properties of the first derivative.

Assignment
This week’s mastery quiz asks you to solve a system of equations and graph a piecewise-defined function. The
steps involved in solving this week’s problem will help you with the questions on the mastery quiz.
Project 1 will be due at the beginning of next week’s lab.

The Problem: Design a Roller Coaster
Suppose we are asked to design a simple ascent and drop roller coaster with an overall horizontal displacement
of 200 feet. By studying pictures of our favorite roller coasters, we decide to create our roller coaster using a line,
a parabola and a cubic. We begin the ascent along a line y = f1(x) of slope 1.5 for the first 20ft horizontally.
We continue the ascent and begin the drop along a parabola y = f2(x) = ax2 + bx + c for the next 100ft hori-
zontally. Finally, we begin a soft landing at 30ft above the ground along a cubic y = f3(x) = dx3 +ex2 +fx+g
for the last 80ft.

Here are our tasks:

1. Find a system of 7 equations with the 7 unknowns ({a,b,c,d,e,f,g}) that will ensure that the track is
smooth at transition points.

2. Solve the equations in (1) to find our functions. (We should get a unique solution as we have the same
number of equations and unknowns.)

3. Plot the graph to see the design.

4. Find the maximum height of the roller coaster.

Solving the Problem
We begin by defining our functions in Maple. If we choose the origin as our starting point, our first function
y = f1(x) is a line of slope 1.5 that passes through (0,0), and we have:
> f1:=1.5*x;

> f2:=a*x∧2+b*x+c;
> f3:=d*x∧3+e*x∧2+f*x+g;
We will also need the first derivatives of our functions. (If you do not see why, you will soon.) So we input:
> df1:=diff(f1,x);

> df2:=diff(f2,x);

> df3:=diff(f3,x);

Since our roller coaster consists of three curves, it can be set up mathematically as a piecewise-defined function:

F (x) =




f1(x), 0 ≤ x ≤ 20
f2(x), 20 < x < 120
f3(x), 120 ≤ x ≤ 200

We assign F in Maple as follows:
> F:=piecewise(x<=20, f1, x>20 and x<120, f2, x>=120 and x<=200, f3);

Obviously, we want F (x) to be continuous (so our passengers do not perish). This means that our functions
should be equal at transition points. So we get the following equations:
> eq1:=eval(f1,x=20)=eval(f2,x=20); % f1(20)=f2(20)
> eq2:=eval(f2,x=120)=eval(f3,x=120); % f2(120)=f3(120)

If we are to have a smooth track, we cannot have abrupt changes in direction, so the first derivative F ′(x)
should also be continuous. That is, the first derivatives of our functions should also be equal at transition
points. So we get:
> eq3:=eval(df1,x=20)=eval(df2,x=20); % f1’(20)=f2’(20)
> eq4:=eval(df2,x=120)=eval(df3,x=120); % f2’(120)=f3’(120)

To start our landing at 30ft above the ground for the last 80ft, we would have:
> eq5:=eval(f3,x=120)=30; % f3(120)=30

Finally, in order to have a soft landing, the track should be tangent to the ground at the end:
> eq6:=eval(f3,x=200)=0; % f3(200)=0
> eq7:=eval(df3,x=200)=0; % f3’(200)=0

We now have a system of 7 equations and 7 unknowns. We solve using the solve command and assign the
solutions as follows:
> values:=solve({eq1,eq2,eq3,eq4,eq5,eq6,eq7} , {a,b,c,d,e,f,g});
We can now determine our functions:
> f2:=eval(f2,values);

> f3:=eval(f3,values);

You can now view and differentiate F (x):
> F:=eval(F,values);

> dF:=diff(F,x);

We can see what our coaster looks like with the following plot command:
> plot(F, x=0..200, y=-50..150, discont=true);

Note: The discont=true option will show us if the graph has any gaps or holes.

To find the maximum height, find where the graph has a horizontal tangent line and evaluate F (x) at each
point. The largest is the maximum height of the coaster.
> solve(dF=0, x);

